

Evaluating Integrals Using Spherical Coordinates

We will begin with plotting using ‘`sphereplot`’ and then address the integrals. In parametric form `sphereplot` uses $[\rho, \theta, \phi]$, where one of the three is in terms of the other two or is a constant. Remember that ϕ is measured down from the “north pole” or the usual z -axis. To understand the relationships between the coordinate systems it is good to start with the r of polar or cylindrical coordinates and recall that

$$x = r \cos \theta$$


$$y = r \sin \theta$$

Building on this, use the fact that $r = \rho \sin \phi$ to obtain for $[\rho, \theta, \phi]$


$$\begin{array}{ll} x = (r) \cos \theta & x = \rho \sin \phi \cos \theta \\ y = (r) \sin \theta & \implies y = \rho \sin \phi \sin \theta \\ z = \rho \cos \phi & z = \rho \cos \phi \end{array}$$

It helps if we understand what it means for each of the variables to be held constant while the other two vary. In the first of three simple plots we hold ρ constant. This should produce some portion of a sphere. In order to have the axes appear at the origin we have included a graph in white which we do not list here. It is not necessary to use the Greek names of the variables, but for demonstration purposes it is easiest. The output is on the left.

```
> sphereplot([2,theta,phi],theta=0..Pi/2,phi=0..Pi/2,color=green);
```



ρ is constant




θ is constant

With the output above on the right, hold θ constant. The surface should be in the vertical plane $\theta = \pi/3$.

```
> sphereplot([rho,Pi/3,phi],rho=1..2,phi=Pi/4..Pi/2,color=blue);
```

Now hold ϕ constant. This restricts the surface to a portion of some cone.


```
> sphereplot([rho,theta,Pi/3],rho=1..2,theta=0..Pi/2,color=blue);
```



ϕ is constant

Recall that in polar coordinates $r = a \sec \theta$ produced a vertical line, $x = a$. And, $r = a \csc \theta$ produced a horizontal line, $y = a$. The equation $r = a \cos \theta$ yielded a circle, but here, the angle ϕ is measured from the z -axis, so $\rho = a \cos \phi$ would produce a circle on the xy -plane for *every* θ . This results in a sphere tangent to the xy -plane. Let's combine that with an equation of the form $\rho = a \csc \phi$, which is a vertical cylinder of radius a .


```
> A:=sphereplot([4*cos(phi),theta,phi],theta=0..2*Pi,phi=0..Pi/4,color=cyan):
> B:=sphereplot([2*csc(phi),theta,phi],theta=0..2*Pi,phi=Pi/4..Pi/2):
> display(A,B);
```



Now we turn to the integrals and begin with an example.

Example 1 Find the volume of the solid inside the sphere $x^2 + y^2 + z^2 = 9$, above the xy -plane, below the upper portion of the cone $x^2 + y^2 = z^2$, and excluding the first octant. This solid would require more than one integral in rectangular coordinates, while in spherical it is quite straightforward. Remember that in spherical coordinates $dV = \rho^2 \sin \phi \, d\rho \, d\phi \, d\theta$.

```
> with(student): with(plots):
> sph1:=sphereplot([3,theta,phi],theta=Pi/2..2*Pi,phi=Pi/4..Pi/2,color=blue):
> cone1:=sphereplot([rho,theta,Pi/4], theta=Pi/2..2*Pi,rho=0..3,color=red):
> side1:=sphereplot([rho,Pi/2,phi],rho=0..3,phi=Pi/4..Pi/2,color=cyan):
> side2:=sphereplot([rho,0,phi],rho=0..3,phi=Pi/4..Pi/2,color=cyan):
> display(sph1,cone1,side1,side2);
```



```
> V1:=Tripleint(rho^2*sin(phi),rho=0..3,phi=Pi/4..Pi/2,theta=Pi/2..2*Pi);
```

$$V1 := \int_{\pi/2}^{2\pi} \int_{\pi/4}^{\pi/2} \int_0^3 \rho^2 \sin(\phi) d\rho d\phi d\theta$$


```
> valV1:=value(V1);
```

$$valV1 := \frac{27}{4}\pi\sqrt{2}$$

Example 2 Find the volume of the solid inside the sphere $x^2 + y^2 + z^2 = 4$ that is **outside** the cylinder $x^2 + y^2 = 1$.

A three-quarter view of this solid results from:

```
> sph2:=sphereplot([2,theta,phi],theta=Pi/2..2*Pi,phi=Pi/6..5*Pi/6,color=blue):
> cyl2:=sphereplot([csc(phi),theta,phi],theta=Pi/2..2*Pi,phi=Pi/6..5*Pi/6,color=red):
> side3:=sphereplot([rho,0,phi],rho=csc(phi)..2,phi=Pi/6..5*Pi/6,color=cyan):
> side4:=sphereplot([rho,Pi/2,phi],rho=csc(phi)..2,phi=Pi/6..5*Pi/6,color=cyan):
> display(sph2,cyl2,side3,side4);
```



Integrate first with respect to ρ and imagine generating rays emanating from the origin which begin as they reach the cylinder and end as they exit through the sphere. Then add up those rays in vertical planes that contain the z -axis to form little “orange slices” by integrating with respect to ϕ from $\pi/6$ down to $5\pi/6$. End by integrating with respect to θ from the “first slice” at $\theta = 0$ and around to $\theta = 2\pi$.

$$\int_0^{2\pi} \int_{\pi/6}^{5\pi/6} \int_{\csc \phi}^2 \rho^2 \sin(\phi) d\rho d\phi d\theta = 4\pi\sqrt{3}$$

C3M12b Problems In problems 1 and 2, plot the solid and find its volume using spherical coordinates and Maple.

1. Q is bounded by the cone $z = \sqrt{x^2 + y^2}$ and the plane $z = 3$. A horizontal plane in spherical coordinates has the form $\rho = a \sec \phi$.
2. R is the solid that lies above the cone $z = \sqrt{x^2 + y^2}$ and inside the sphere $x^2 + y^2 + z^2 = 6z$.
3. Evaluate the integral by changing to spherical coordinates:

$$\int_0^{\sqrt{2}} \int_y^{\sqrt{4-y^2}} \int_0^{\sqrt{4-x^2-y^2}} \sqrt{x^2+y^2+z^2} \, dz \, dx \, dy$$